
Chapter 2

Evaluators
& Dynamic Semantics

Semantics is the study ofmeaning, just as syntax is the study of structure. If you
know the grammar of a language, you can tell whether a sentence is syntacti-
cally correct, and what its structure is. To understand its meaning, if there is
any, you must know the semantics of the language.

‘Meaning’ can takemany forms. In simple cases, wemay be interested only
in knowing what output we will get if we run a programwith some particular
input – exactly how we arrive at that output is uninteresting. In other cases,
we may be interested in knowing exactly what steps are executed, and in what
order. Or, wemaywant to knowwhat armmovements a robot will makewhen
we run its program. A good semantics description gives us exact, unambigu-
ous answers to such questions, without overspecifying uninteresting details.

For example, the specification of C will tell you details of when you can be
sure that values are written tomemory, but leaves out other details, such as the
evaluation order of expressions and the size and exact semantics of integers.
The definition of a functional language will give rules for computing the value
of expressions, and maybe also for how and when expressions are evaluated.

2.1 Dynamic and Static Semantics

For programming languages, we typically distinguish between static and dy-
namic semantics.

What is classified as static or dynamic depends a bit on the language, but
in general, static semantics (Chapter 3) deals with determining whether a pro-
grammakes enough sense to be executed at all, while dynamic semantics deals
with what happens when you execute a program.

Static semantics may include some rules, such as typing rules, that could
have been part of the grammar, but was left out to keep the grammar context-
free. For example, that the types of the operandsmustmatch the declaration of
an operator, or that a variable must be declared before it is used. Static seman-
tics may also specify how the uses of names are resolved, and how overloaded
names are dealt with.

2. Evaluators & Dynamic Semantics A.H. Bagge

1 module syn::SimplTrivial
extend syn::SimplLexical;

3

start syntax Program = Expr expr;
5

syntax Expr
7 = INT num
| bracket "(" Expr e ")"

9 > left (Expr e1 "*" Expr e2 | Expr e1 "/" Expr e2)
> left (Expr e1 "+" Expr e2 | Expr e1 "-" Expr e2)

11 > left (Expr e1 "==" Expr e2 | Expr e1 "\<" Expr e2)
| "if" Expr cond "then" e1 Expr "else" Expr e2 "end"

13 ;

Listing 2.1: Evaluator/src/syn/SimplTrivial.rsc: Syntax for Trivl – a trivial expression
language without variables or functions.

Dynamic semantics specifies what a program does – what its results are or
what actions it performs. This may also entail some of what is listed under
static semantics above, particularly in dynamic languages.

In this chapter, we will discuss dynamic semantics. We start by defining
the semantics of simple programs that just return a value. Later we will also
consider programs that work on state.

2.2 The Simpl-Exp Expression Language

We can specify semantics formally, using a specification formalism, or infor-
mally, using a natural language such as English. Another approach is to just
give an implementation of a language, and say that the implementation is also
the specification. This is often the case for languages that are under develop-
ment, or for which there is only a single implementation. If the implementa-
tion is clear and readable, this canmake a lot of sense, and gives a fairly decent
specification – with the additional benefit that we can test our assumptions
against an executable implementation.

For a widely adopted language with multiple implementations, there’s a
greater need for a proper specification, so that developers and language im-
plementers can agree on the exact meaning of programs.

For us, simplicity andunderstanding the fundamental concepts of languages
is themost important thing, sowe’ll startwith the implementation-as-specification
approach, and develop an evaluator in Rascal for a series of increasingly com-
plex languages.

2.2.1 Evaluating Trivial Expressions

Let’s start by considering the trivial expression language in Listing 2.1, and
develop an evaluator for it.

The syntax only provides for numbers, arithmetic expressions and condi-
tions. Valid programs include expressions such as 1 + 2 * 3, (2 * 3) + 7,

INF225 Notes 2.2. The Simpl-Exp Expression Language

Concrete Syntax and Pattern Matching in Rascal

Concrete Syntax
Rascal allows you to write parse trees and
patterns using the concrete syntax of the
object language (the language we’re defin-
ing). Concrete syntax fragments are en-
closed in back-quotes (‘...‘), and preceded
by the name of the appropriate non-terminal
in parenthesis:

(Expr)‘a + 2‘

Pattern variables in concrete syntax frag-
ments are enclosed in angle brackets (<...>).
The variables should be typedwith the name
of the corresponding non-terminal, in order
to help out the Rascal parser:

(Expr)‘<Expr b> + 2‘

Variables such as b above are also calledmeta
variables – they are variables in the host lan-
guage, rather than the object language (like
the variable a in the former code fragment).

Pattern Matching
Concrete syntax can be used in pattern
matching. For example:

if((Expr)‘<Expr e1> + 0‘ := e) {
e = e1;
}

On a successful match, the pattern variables
are bound to the corresponding subtrees of
the parse tree matched against.

You can also use concrete syntax patterns
a the function parameter list:

str term((Expr)‘<Expr a>+<Expr b>‘) {
// <> can be used inside strings to

// ’ interpolate ’ values into the string .
// They can contain any expression , incl
// function calls
return "Add(<term(a)>, <term(b)>)";
}

If you use pattern matching in parameter
lists, it’s a good idea to provide a default case
that catches calls that don’t match any of the
patterns:

default str term(Expr e) {
throw "unknown expression <e>";
}

To obtain the string corresponding to a
parse tree (i.e., the input string before pars-
ing), use the unparse function:

rascal> (Expr)‘a + b‘
sort("Expr"): ‘a + b‘
Tree: appl(prod(sort("Expr"),...
rascal> unparse((Expr)‘a + b‘)
str: "a + b"

15



2. Evaluators & Dynamic Semantics A.H. Bagge

and if 2 > 3 then 4 else 5 end. We’re of course free to assign any kind
of meaning to our language, but to avoid surprising our users, we’ll stick to a
fairly conventional interpretation: A program yields a value, computed by ap-
plying standard integer arithmetic. For example, the value of the expression
(2 * 3) + 7 is 13.

First, we need to define what a value is. We’ll make a Value type, which
we’ll use in our evaluator. Our language only supports integers, so our value
type has only a single Int case. In Rascal, this gives us the data type:

8 data Value = Int(int intValue); Evaluator/src/eval/EvalTrivial.rsc

To assign meaning to expressions, we define an eval function, with one
case for each kind of expression. The function should accept an expression
and return its value, giving us the signature:

1 public Value eval(Expr e);

2.2.2 The Trivial Evaluator

We now define evaluation functions for each language construct.
Integers: The value of an integer literal i (matched by the concrete syntax frag-
ment ‘<INT i>‘), is its integer value:

11 public Value eval((Expr)‘<INT i>‘) {
12 return Int(toInt(unparse(i)));
} Evaluator/src/eval/EvalTrivial.rsc

We use unparse to make a string from a parse tree, then toInt to convert that
string into an integer. The actual integer is wrapped in an Int constructor, in
accordance with our Value data type.
Arithmetic: In the case of arithmetic expressions, wemust first obtain the value
of the operands, by evaluating them recursively. We can then apply the oper-
ation and wrap up the result. For example, for addition:

25 public Value eval((Expr)‘<Expr e1>+<Expr e2>‘) {
26 return Int(eval(e1).intValue + eval(e2).intValue);
} Evaluator/src/eval/EvalTrivial.rsc

Parentheses: The value of a parenthesis expression is simply the value of the
contained expression:

15 public Value eval((Expr)‘(<Expr e>)‘) {
16 return eval(e);
} Evaluator/src/eval/EvalTrivial.rsc

Comparison: When it comes to comparison operators and conditionals, wemust
decide how to handle Boolean values, as our language so far only deals with
integers. A convenient approach is that of C, where 0 corresponds to false,
and any non-zero value is true. So, to compare two values, we first evaluate
them, then do the comparison, and then select either 0 or 1 (any non-zero value
would work) as the return value:

16



INF225 Notes 2.2. The Simpl-Exp Expression Language

40 public Value eval((Expr)‘<Expr e1>==<Expr e2>‘) {
return Int(eval(e1).intValue == eval(e2).intValue ? 1 : 0);

42 } Evaluator/src/eval/EvalTrivial.rsc

Conditionals: For if expressions, we have to choose between two expressions,
based on the value of the condition. We must be careful and only evaluate
one of the branches. At best, evaluating both branches will result in somewhat
slower execution time; worst case, one of the branches may contain code that
fails in some way, or doesn’t terminate. The code for if is:

51 public Value eval((Expr)
52 ‘if <Expr c> then <Expr e1> else <Expr e2> end‘) {

if(eval(c).intValue != 0)
54 return eval(e1);
else

56 return eval(e2);
} Evaluator/src/eval/EvalTrivial.rsc

Programs: Finally, we provide an evaluator for programs – which is simple
since a program is just an expression:

59 public Value eval((Program)‘<Expr e>‘) {
60 return eval(e);
} Evaluator/src/eval/EvalTrivial.rsc

We can now try the evaluator on a few simple programs:

1 rascal>eval((Program)‘1+2*3‘)
Value: Int(7)

3

rascal>eval((Program)‘(1+2)*3‘)
5 Value: Int(9)

7 rascal>eval((Program)‘if 1 then 2 else 1/0 end‘)
Value: Int(2)

9

rascal>eval((Program)‘if 0 then 2 else 1/0 end‘)
11 |rascal://eval::EvalTrivial|(1041,2,<36,37>,<36,39>):

ArithmeticException("/ by zero")

Questions

1. Can you think of any drawbacks to specifying a language in this way,
compared to using plain English or a mathematical formalism? What
about advantages?

2. Let’s say we implemented the evaluator in Java instead; using objects
with an eval method as nodes in an abstract syntax tree representation
of programs. For example, the tree node for addition would have this
evalmethod:

2. Evaluators & Dynamic Semantics A.H. Bagge

1 module syn::SimplVars
2 extend syn::SimplLexical;

4 start syntax Program = Expr expr;

6 syntax Expr
= ID var

8 | INT num
| bracket "(" Expr e ")"

10 > left (Expr e1 "*" Expr e2 | Expr e1 "/" Expr e2)
> left (Expr e1 "+" Expr e2 | Expr e1 "-" Expr e2)

12 > left (Expr e1 "==" Expr e2 | Expr e1 "\<" Expr e2)
| "if" Expr cond "then" e1 Expr "else" Expr e2 "end"

14 | "let" ID var "=" Expr e1 "in" Expr e2 "end"
;

Listing 2.2: Evaluator/src/syn/SimplVars.rsc: Syntax for SimplVars – a trivial expres-
sion language with variables.

1 public Value eval() {
2 return new IntValue(e1.eval().intValue()

+ e2.eval().intValue());
4 }

Would this change the semantics in any way? (Hint: is there a difference
between integers in Java and Rascal?)

2.2.3 Variables and Environments

The language in the previous section has the power of a very simple calcu-
lator, and is not very exciting. The next step is to add variables and variable
declarations.

First, we will set some simple rules for variables:

• Variables are bound to values in a let-expression.

• The binding of a variable in a let-expression has no effect outside that
expression.

• We track the currently bound variables in an environment.

• The value of a variable expression is the value of the variable in the en-
vironment.

• Accessing a variable that is not bound is illegal.

Next, we need to add productions to the grammar for defining and access-
ing variables. The extended grammar is shown in Listing 2.2.

Then, we must decide how to represent the environment. Fundamentally,
its purpose is to keep track of mappings from variables to values, so we may
represent it as a map from strings (variable names) to Values:

INF225 Notes 2.2. The Simpl-Exp Expression Language

8 alias Env = map[str, Value]; Evaluator/src/eval/EvalVars.rsc

The environment is only used when defining and using variables. How-
ever, since other expressions may have variables uses as sub-expressions, we
need to track the environment when processing other kinds of expressions as
well. We modify the eval function to accept the current environment as a pa-
rameter:

1 public Value eval(Expr e, Env env);

We must then modify all the eval cases to accept the extra argument, and
pass it on to recursive calls. For example, for addition:

33 public Value eval((Expr)‘<Expr e1>+<Expr e2>‘, Env env) {
34 return Int(eval(e1, env).intValue + eval(e2, env).intValue);
} Evaluator/src/eval/EvalVars.rsc

We are now ready to tackle the variables themselves. To obtain the value
of a variable, we look it up in the environment:

19 public Value eval((Expr)‘<ID x>‘, Env env) {
20 return env[unparse(x)];
} Evaluator/src/eval/EvalVars.rsc

Again, we use unparse to obtain a string from the parse tree matched in the
concrete syntax pattern ‘<ID x>‘. We should also include error handling code,
but we can postpone this, and rely on Rascal to throw an exception if the vari-
able is not defined in the environment map.

The let-expression is slightly more complicated, but not much. It is used
like this:

1 let x = 5+2 in x * x end

In the expression above, the variable x should be bound to the value of 5+2
while evaluating x * x. We implement this as follows:

70 public Value eval((Expr)
‘let <ID x> = <Expr e1> in <Expr e2> end‘, Env env) {

72 v = eval(e1, env);
return eval(e2, env + (unparse(x) : v));

74 } Evaluator/src/eval/EvalVars.rsc

First, we evaluate the first expression (e1), obtaining a value v. Then we eval-
uate and return the value of the second expression (e2), in an environment to
which we have added the binding of the variable identified by x to v.

Questions

1. What happens if we mention the same variable name we’re defining in
the initializer for that variable? For example:

1 ... let x = 5+x in x * x end ...

2. What happens if we redefine a variable which is already defined? For
example:

2. Evaluators & Dynamic Semantics A.H. Bagge

1 let x = 1 in let x = 5 in x * x end end

(To answer this, you need to know the semantics of Rascal maps.)

2.3 Environmental Concerns

Our language is still to simple to do useful computation, but we can use it to
illustrate one important concept: the environment.

With the possible exception of constant expressions, the evaluation of any
part of a program will depend on information from the surrounding environ-
ment. For example, an expression such as x + 5 is meaningless on its own,
as we don’t know what x is. If we see the wider context of the expression, for
example,

1 let x = 7 in
2 x + 5
end

we can construct an environment x 7→ 7 and find the value of x + 5.
Environments are used in both language specification and implementation.

Compiler literature typically uses the term symbol table.
The environment is used to keep track of contextual information such as

the bindings of variables, and possibly other information, like the name of
the current function or class, information about language options, and so on,
depending on the language. In its simplest form, as in the previous section,
the environment is simply a map from variables to values.

2.3.1 Environment Interface

As we proceed with more advanced languages, it is useful to abstract away
the details of how the environment is implemented, and focus just on how it
is used. Our use of the environment in Section 2.2.3 can be summed up as the
following operations:

• env’ = declare(x, v, env) – binding a new variable x to the value v
in env, overriding any previous binding in env. The result is a new envi-
ronment env’.

• v = lookup(x, env) – look up a variable x in an environment env, yield-
ing its binding v. The result is undefined if x is not bound in env.

We also need a way to construct new environments, and check whether a vari-
able is bound:

• env = newEnv() – create a fresh, empty environment env

• isDefined(x, env) – true if the variable x is bound in the environmentenv

For now, we’ll store data values in the environment; later, we’ll also use it bind
variables to function definitions, types and storage locations.

We assume that values are immutable in our implementation language (as
they are in Rascal), which is why declare returns a new environment.

The Rascal declarations for the above operations are shown in Fig. 2.1. Note
that we’ve parameterized the Env typewith the type of values we’ll store in the

INF225 Notes 2.3. Environmental Concerns

public Env[&T] declare(Tree name, &T val, Env[&T] env)

public &T lookup(Tree name, Env[&T] env)

public Env[&T] newEnv(type[&T] _)

public bool isDefined(Tree name, Env[&T] env)

Figure 2.1: Rascal declarations for the environment operations. &T indicates a
type parameter (can be any type). Tree is the type of parse trees.

environment. We also use the parse trees of variable identifiers as keys in the
environment, instead of strings. This will allow us to more easily use struc-
tured names later on, and also makes it easy to provide good error messages,
since parse trees carry location information.

2.3.2 Semantics of the Environment

Sincewe’re dealingwith semantics of languages, we should keep all our ducks
in a row, and not leave the semantics of fundamental concepts like the environ-
ment unspecified. Below, we provide equational axioms that define the impor-
tant characteristics of the environment operations. In the axiom, xwill be any
variable, v any value, and e any environment.
New Environment. Nothing is defined in an empty environment, so isDefined
always returns false. The lookup operation is undefined on empty environ-
ments (we may implement it in practice using exceptions, for example):

isDefined(x,newEnv()) ⇐⇒ false (2.1)
lookup(x,newEnv()) ⇐⇒ undefined (2.2)

Binding a Variable. Adding a variable to the environment, and then looking up
the same variable, yields the same value. Similarly for isDefined.

lookup(x,declare(x, v, e)) ⇐⇒ v (2.3)
isDefined(x,declare(x, v, e)) ⇐⇒ true (2.4)

This also implies that new bindings override old ones with the same name.
Binding Unrelated Variables. Adding a variable x2 doesn’t affect the binding of
the variable x1, as long as x1 6= x2 – i.e., we get the same result from looking
up x1 whether or not we have declared some other variable x2.

lookup(x1,declare(x2, v, e)) ⇐⇒ lookup(x1, e), for x1 6= x2 (2.5)
defined(x1,declare(x2, v, e)) ⇐⇒ defined(x1, e), for x1 6= x2 (2.6)

2.3.3 Namespaces

We may also want to have separate namespaces for different kinds of names,
such type names, variable names, module names or function names. For ex-
ample, if we allow user-defined types, we could allow types and variables to

2. Evaluators & Dynamic Semantics A.H. Bagge

share the same name, but have separate definitions. In this case, we could use
one environment for each namespace, or add an extra parameter to the opera-
tions, selecting what kind of name we’re working with.

Languages like C++ and Java support named scopes (e.g., java.lang.io) –
this complicates things quite a lot, and requires a more advanced environment
structure.

2.4 Functions

Let us now turn our attention back to our little language, and add one of the
features it’ll need to make it in the real world: functions.

Functions have two key features that make them important: they abstract,
so that we don’t need to know the details of how something is done in order to
use the results; and they provide parametrization, so that we can use the same
piece of code to do many different useful computations. Additionally, having
recursive functions will allow us to do arbitraty computation, even without
have loop constructs in the language.First-class values can be passed

as arguments, returned,
assigned to a variables and
generally be used the same way
other values.

In some languages functions are special entities that are declared andnamed,
but can’t be passed as arguments, stored in variables or returned from func-
tions. This is true for Javamethods, for example. Other languages, such as Lisp
and Haskell, treat functions as first-class values, which can be passed around
like any other value.

In our language, which is (so far) expression-oriented, the latter approach
makes sense: we will treat functions as values. Functions-as-values is nor-Lambda expressions are

functions that can be defined
and used without being bound
to an identifier.

mally combined with allowing anonymous functions, also known as lambda
expressions. First class functions and lambda expressions are very popular on
the interwebs, so these features should help our new language on its way to
success.

With functions being values, we can reuse our existing let construct to
define functions. For example:

1 let f = fun x => x * x in
f(2)

3 end

A function expression is introduced by the ‘fun’ keyword, followed by the list
of parameters, a ‘=>’ sign, and the body of the function (an expression).

The updated syntax is shown in Listing 2.3. In addition to declaring func-
tions with let, we can also use them directly in a call – though this is usually
less useful:

1 (fun x => x * x)(5)

2.4.1 Functions in the Evaluator

Having extended the syntax, we now need to define the semantics of function
declaration and calls by extending the evaluator.

Function Values. First of all, we need to extend our idea of what a value is to
include functions:

1 module syn::SimplFuns
extend syn::SimplLexical;

3

start syntax Program = Expr expr;
5

syntax Expr
7 = ID var
| INT num

9 | bracket "(" Expr e ")"
| Expr fun "(" {Expr ","}* args ")"

11 > left (Expr e1 "*" Expr e2 | Expr e1 "/" Expr e2)
> left (Expr e1 "+" Expr e2 | Expr e1 "-" Expr e2)

13 > left (Expr e1 "==" Expr e2 | Expr e1 "\<" Expr e2)
| "if" Expr cond "then" e1 Expr "else" Expr e2 "end"

15 | "let" ID var "=" Expr e1 "in" Expr e2 "end"
> "fun" {ID ","}* params "=\>" Expr body

17 ;

Listing 2.3: Evaluator/src/syn/SimplFuns.rsc: Syntax for SimplFuns – a simple ex-
pression language with functions. The new syntax rules are on lines 10 (func-
tion calls) and 16 (functions).

10 data Value
= Int(int intValue)

12 | Fun(list[ID] params, Expr body)
; Evaluator/src/eval/EvalFuns.rsc

The new case for functions include a parameter list, as a list of identifiers, and
a body – an expression. Both ID and Expr are parse tree types, referring to
non-terminals in our grammar.

We also need to tell the evaluator how to handle this kind of value:

38 public Value eval((Expr)‘fun <{ID ","}* params> =\> <Expr body>‘,
Env[Value] env) {

40 return Fun([p | p <- params], body);
} Evaluator/src/eval/EvalFuns.rsc

The concrete syntax match expression is slightly more complicated than we’ve
seen before. Essentially, <{ID ","}* params>means “match a comma-separated
list of IDs, and bind to the variable params”. In the returned expression, we
use [p | p <- params] (build a list of p’s, with one p for each p in params)
– this is just a simple way of converting a comma-separated list of identifiers
into a plain list of identifiers.

For example, we can evaluate a function and obtain a function value like
this (we’ve defined an extra function printlnValue to print values prettily,
otherwise we’ll get huge parsetrees in the output):

1 rascal> printlnValue(eval((Expr)‘fun x, y, z =\> x * y + z‘,
2 newEnv()))

Fun([(ID)‘x‘, (ID)‘y‘, (ID)‘z‘], (Expr)‘x * y + z‘)
4 ok

FunctionDeclarations. Wehave no separate function declaration construct (yet);
we just use the same let as before. Given our above code, the old code for let
should still work. Here’s the code again, updated to use the new environment
operations:

106 public Value eval((Expr)
‘let <ID x> = <Expr e1> in <Expr e2> end‘, Env[Value] env) {

108 v = eval(e1, env);
return eval(e2, declare(x, v, env));

110 } Evaluator/src/eval/EvalFuns.rsc

Notice how the evaluation of let starts by evaluating the expression e1. This
would be the function in the case of a function declaration. This is why we
need to have an evaluation function for function values, even though it hardly
does any work.
Function Calls. Now for the interesting stuff. The evaluation of a function call
consists of four steps:

1. Evaluate the function expression, to get the function value

2. Evaluate the actual arguments

3. Bind the actual arguments to the formal parameters of the function

4. Evaluate the function body in the resulting environment

The Rascal code for this is:

43 public Value eval((Expr)‘<Expr f>(<{Expr ","}* args>)‘,
44 Env[Value] env) {

fVal = eval(f, env); // evaluate function expression
46 if(Fun(params, body) := fVal) {

fEnv = env; // set up environment
48

for(<arg, param> <- zip([a | a <- args], params)) {
50 argVal = eval(arg, env); // evaluate argument

fEnv = declare(param, argVal, fEnv); // bind parameter
52 }

54 return eval(body, fEnv); // evaluate body in fEnv
}

56 else throw "Calling a non-function: <unparse(f)>, at <f@\loc>";
} Evaluator/src/eval/EvalFuns.rsc

Wehave combined step 2 and 3 into one pass over the argument and parameter
lists (zipped together into a list of pairs). We’ve also added a little bit of error
checking by complaining if the thing we’re trying to call is not a function (i.e.,
evaluates to something other than a function value), but further error checking
(such as checking whether the arguments match the parameter list) has been
omitted. Rascal will however throw its own exceptions in the case of errors.

Example 1: Step-by-Step Function Evaluation
Consider the following code:

1 let f = fun x, y, z => x * y + z in
f(1, 2, 3)

3 end

Its evaluation will proceed as follows, starting with evaluating the outer let
expression:

1. First, the function expression is evaluated, yielding a function value Fun([(ID)‘x‘, (ID)‘y‘, (ID)‘z‘],
(Expr)‘x * y + z‘)

2. Then, the variable f is bound to the function value.

3. Finally, the body of the let, containing the function call is evaluated:

a) First, we evalute the function expression – f, which yields the func-
tion value above.

b) Then,we evaluate each argument, yielding a series of values Int(1),
Int(2), Int(3), and bind them to the parameters; x 7→ Int(1), y 7→
Int(2), z 7→ Int(3).

c) Finally, we evaluate the function body in the environment produced
by the previous step. This will produce the final result, Int(5) (1 ∗
2+ 3 = 5).

Exercises

1. Evaluate the following programs by hand:

a)1 let f = fun x, y => x * y in
2 f(f(2, 2), 3)
end

b)1 let fac = fun n => if n < 1 then 1 else n*fac(n-1) end in
fac(3)

3 end

c)1 let x = 5 in
let f = fun a => a * x in

3 f(10)
end

5 end

d)1 let x = 5 in
let f = fun a => a * x in

3 let x = 7 in
f(10)

5 end
end

7 end

Study the last two programs carefully, consulting our definition of func-
tion calls. Is the answer in d) different from that in c)? Which value is the
x inside the function bound to in each case?

2.5 Scoping

If you paid careful attention to the exercise at the end of the previous section,A free variable is one that is not
bound (by let or as a
parameter) in a certain context.
E.g., x is free in x * 2, but
bound in let x = 4 in x * 2
end.

youwill have noticed that the binding of free variables in function bodies is de-
termined by the caller’s environment, rather than by the environment that was
in place at the point where the function was defined. The example program
below makes this apparent:

1 let f = fun => x * x in
2 let x = 2 in

f()
4 end end // result is 4

There is no variable x at the point where the function f is defined. Instead, x
refers to whatever variable x is active when the function is called. This is called
dynamic scoping.

Another approach is to say that identifiers refer to whatever names are ac-
cessible at the point where the function is defined. This wouldmake the above
program invalid – since x is undefined when f is being defined. Instead, we
need to add an outer definition of x. For example:

1 let x = 4 in
2 let f = fun => x * x in

let x = 2 in
4 f()
end end end // result is 16

Here, the inner definition of x has no effect, as the x in f refers to the outermost
x = 4. This is called lexical or static scoping.

2.5.1 Scope Terminology

A scope is a collection of identifier bindings – i.e., what is captured by the en-
vironment at some point in the code or in time.

The scope of a declaration includes all the points in the code where the de-
clared identifier is bound according to the declaration. In lexical scoping, the
scope of a declaration is typically either textually included in the declaration
construct, or extends to the nearest scoping container. This is illustrated in
Fig. 2.2, where the scopes of each variable declaration is marked in the code
margin. In dynamic scoping, the scope of a declaration is determined at run-
time, and lasts until the program exits from the let or other scoping construct.

Bindings can also be shadowed by declaring a new variable with the same
name as an in-scope variable. This results in the outer variable being tem-
porarily out of scope while the inner variable is in scope (though, it may still
be possible to access the shadowed variable, e.g., through qualified names).
The shadowed variable still exists and is active – it is only inaccessible due to
its name being out of scope. This forbidden by some languages, such as Java,
as it may sometimes lead to confusion.

An identifier is said to be in scope at a given point in the code if it occurs
within the scope of its declaration.

1 let x = 42 in
2 x // scope of x

// ...
4 end

1 int x;
2 x {

int y,
4 y z = y;

// ...
6 }

// ...

1 let x = 4 in
x let y = 2 in

3 y let x = y in
x x

5 end
end

7 end

Figure 2.2: Examples of the scopes of declarations. The scope of a let (left) is
the body of the let. The scope of Java variable declarations (middle) extend
from immediately after the new variable is mentioned, to the end of the cur-
rent block. Inner declarations may shadow outer declarations, putting some
identifiers temporarily out of scope (right).

2.5.2 Dynamic Scoping

Dynamic scoping is seldom used nowadays, as it tends to lead to confusing
programs. With dynamic scoping, you can’t fully understand what a piece of
code does without also looking at context in which it is used. This is particu-
larly problematic if functions can be defined and rebound on the fly.

In an evaluator-based approach such as ours, dynamic scoping is very easy
to implement – it’s what you end upwith if you don’t do anything in particular
to deal with scoping.

• For dynamic scoping, we only need to know the environment that is in
place at the time a function is called. There is no need to know anything
about the environment as it was when the function was called.

• Thus, our representation of functions from Section 2.4.1 is sufficient:

12 | Fun(list[ID] params, Expr body) Evaluator/src/eval/EvalFuns.rsc

• Emacs Lisp, and some older versions of Lisp use dynamic scoping.

2.5.3 Lexical Scoping

With lexical scoping, the binding of an identifier follows from the surrounding
program text – i.e., the lexical context. This is typically clearer for the program-
mer, and it also allows us to statically check whether a program uses variables
that haven’t been declared.

Lexical scoping complicates the dynamic semantics somewhat. We can no
longer evaluate a function call without knowing what was in scope at the time
the function was declared. For this, we need to store the definition environment
of the function together with its body, so we can later restore the environment.
This means extending the definition of function values:

10 data Value
= Int(int intValue)

12 | Fun(list[ID] params, Expr body, Env[Value] env)

; Evaluator/src/eval/EvalLex.rsc

Function declarations. Next, we must extend the evaluation of function expres-
sions to save the environment in the function value:

38 Value eval((Expr)‘fun <{ID ","}* params> =\> <Expr body>‘,
Env[Value] env) {

40 return Fun([p | p <- params], body, env);
} Evaluator/src/eval/EvalLex.rsc

Function calls. Finally, the evaluation of function calls must be changed, so that
the function body gets evaluated in the function’s own environment:

43 Value eval((Expr)‘<Expr f>(<{Expr ","}* args>)‘, Env[Value] env) {
44 fVal = eval(f, env); // evaluate function expression

if(fVal is Fun) {
46 fEnv = fVal.env; // function environment

for(<arg, param> <- zip([a | a <- args], fVal.params)) {
48 argVal = eval(arg, env); // evaluate argument

fEnv = declare(param, argVal, fEnv); // bind parameter
50 }

52 return eval(fVal.body, fEnv); // evaluate body in fEnv
}

54 else throw "Calling a non-function: <unparse(f)>, at <f@\loc>";
} Evaluator/src/eval/EvalLex.rsc

Themain change compared to the dynamic scoping version of function calls, is
that we now use a separate fEnv for evaluating the body. This environment is
initialised with the saved declaration environment of the function (fVal.env).
As before, the arguments are evaluated in the caller’s environment (env) and
bound in the callee environment (fEnv).

Recursion. If you study the rules for function values and declarations carefully,
you will see that the env that gets added to the function value doesn’t contain
the binding of the function itself. For example, in the following piece of code,
we’ll evalute fun a => f(a) first, resulting in a function value capturing the
environment as it is before we bind the f:

1 let f = fun a => f(a) in ... end

This means that f is not bound within its own body.
An easy solution to this problem is to just add the binding of the function

to the environment while evaluating the call:

58 Value eval((Expr)‘<Expr f>(<{Expr ","}* args>)‘, Env[Value] env) {
fVal = eval(f, env); // evaluate function expression

60 if(fVal is Fun) {
fEnv = fVal.env; // function environment

62 if((Expr)‘<ID fName>‘ := f) { // check if f is a name
fEnv = declare(fName, fVal, fEnv); // for recursive calls

64 }

for(<arg, param> <- zip([a | a <- args], fVal.params)) {
66 argVal = eval(arg, env); // evaluate argument

fEnv = declare(param, argVal, fEnv); // bind parameter
68 }

70 return eval(fVal.body, fEnv); // evaluate body in fEnv
}

72 else throw "Calling a non-function: <unparse(f)>, at <f@\loc>";
} Evaluator/src/eval/EvalLex.rsc

This works nicely for simple recursion, but fails for mutual recursive, e.g., if f
calls gwhich calls f again. We will deal with this problem later.

• We might also mix scope rules; for example, we could make functions
lexically scoped, so that any function call in the body of a function is
resolved at definition time, while still having dynamically scoped vari-
ables.

2.6 Imperative Languages

2.6.1 Store and References

See code from Lecture 11:
https://bitbucket.org/anyahelene/inf225public/src/master/inf225l11?
at=master

See exercise 4A,C,D:
https://bitbucket.org/anyahelene/inf225public/wiki/Exercise%204

2.7 Advanced Scoping

2.7.1 Closures

See exercise 4B:
https://bitbucket.org/anyahelene/inf225public/wiki/Exercise%204

2.7.2 Nested Scopes

See code from Lecture 10:
https://bitbucket.org/anyahelene/inf225public/src/master/inf225l10?
at=master

The Power of the Simpl-Exp Language

You may think that what we’ve made so far
is just a toy, but it is in fact a fairly powerful
programming language – at least if we’re pri-
marily interested in manipulating integers.

Algorithms
Simple algorithms like computing the Fi-
bonacci numbers are easy to implement:

let fib = fun n => if n < 2
then 1
else fib(n-1)+fib(n-2)
end

in
fib(10)
end // result is 89

Encoding Data Structures
We can create simple data structures by en-
coding them as functions that return fields
based on their arguments. For example, a
pair data structure would be a function re-
turning the first field if the argument is 0
and the second field otherwise. We can then
make getters and setters (the setters con-
struct new pairs – we can’t make mutable
data structures) on top of this crude inter-
face:

let pair = fun x, y =>
fun i => if i then y else x end
in
let getX = fun p => p(0) in
let getY = fun p => p(1) in
let setX = fun p, x =>

pair(x, getY(p)) in
let setY = fun p, y =>

pair(getX(p), y) in
getY(setY(pair(17,19),21))
end end end end end // result is 21

This only works with lexical scoping, since
the function returned by pair needs access
to the x and y at the time it was created.

With functions being values, we could
of course also store functions in data struc-
tures. This brings us within reach of object-
orientation and dynamic dispatch (with very
ugly syntax).

This approach is similar to Church encod-
ing, which can be used to encode data struc-
tures, booleans, conditionals and even inte-
gers as functions in lambda calculus.

The Let Construct
The let construct is strictly speaking not
necessary, since functions provide the same
power of binding identifiers. Any let
declaration let x = e1 in e2 end is ex-
actly equivalent to (fun x => e2)(e1) – i.e.,
making a function with the variable as a pa-
rameter and the let-body as the function
body, and then immediately calling it with
the initialiser expression as the argument.

Lambda Calculus
The language we’ve created is closely related
to lambda calculus, which is also the basis of
Lisp, Scheme, and functional programming
languages. The combination of scoping rules
and the ability to create anonymous func-
tions gives us farmore power than onemight
expect.

λ

